

Douven's Model vs Our Bayesian Model of "Abductive Updating"

Branden Fitelson

August 2, 2022

In[*]:= << PrSAT`

Here is a Bayesian model of an agent's priors $cr_{t_0}(\bullet)$, which are such that:

- (1) $cr_{t_0}(\bullet)$ is regular,
- (2) E confirms H : $cr_{t_0}(H | E) > cr_{t_0}(H | \sim E)$,
- (3) **ERA** holds: $cr_{t_0}(H | E \& A) > cr_{t_0}(H | E \& \sim A)$,
 - Here, A is interpreted as asserting that H is the best (or only) explanation of E .
- (4) E is (a priori) irrelevant to A : $cr_{t_0}(A | E) = cr_{t_0}(A | \sim E)$,
- (5) A is (a priori) irrelevant to H : $cr_{t_0}(H | A) = cr_{t_0}(H | \sim A)$, and
- (6) $cr_{t_0}(H) = cr_{t_0}(E) = cr_{t_0}(A) = \frac{1}{2}$.

In[*]:= prior = PrSAT[{
 Pr[H | E] > Pr[H | ~E],
 Pr[H | E & A] > Pr[H | E & ~A],
 Pr[A | E] == Pr[A | ~E],
 Pr[H | A] == Pr[H | ~A],
 Pr[H] == $\frac{1}{2}$, Pr[E] == $\frac{1}{2}$, Pr[A] == $\frac{1}{2}$
 }, Probabilities → Regular, BypassSearch → True]

Out[*]:=

{ {A → {a₂, a₅, a₆, a₈}, E → {a₃, a₅, a₇, a₈},
 H → {a₄, a₆, a₇, a₈}, Ω → {a₁, a₂, a₃, a₄, a₅, a₆, a₇, a₈}},
 {a₁ → $\frac{3}{32}$, a₂ → $\frac{7}{32}$, a₃ → $\frac{5}{32}$, a₄ → $\frac{5}{32}$, a₅ → $\frac{1}{32}$, a₆ → $\frac{1}{32}$, a₇ → $\frac{3}{32}$, a₈ → $\frac{7}{32}$ }}

\mathcal{A}	\mathcal{E}	\mathcal{H}	cr_{t_0}
T	T	T	$\frac{7}{32}$
T	T	F	$\frac{1}{32}$
T	F	T	$\frac{1}{32}$
T	F	F	$\frac{7}{32}$
F	T	T	$\frac{3}{32}$
F	T	F	$\frac{5}{32}$
F	F	T	$\frac{5}{32}$
F	F	F	$\frac{3}{32}$

If we look at this agent's credences over the sub-algebra $\{\mathcal{H}, \mathcal{E}\}$, we have the following (coarse-graining of their) prior.

$In[*]:=$ EvaluateProbability[{Pr[$\mathcal{E} \wedge \mathcal{H}$], Pr[$\mathcal{E} \wedge \neg \mathcal{H}$], Pr[$\neg \mathcal{E} \wedge \mathcal{H}$], Pr[$\neg \mathcal{E} \wedge \neg \mathcal{H}$]}, prior]
 $Out[*]:=$ $\left\{ \frac{5}{16}, \frac{3}{16}, \frac{3}{16}, \frac{5}{16} \right\}$

\mathcal{E}	\mathcal{H}	cr_{t_0}
T	T	$\frac{5}{16}$
T	F	$\frac{3}{16}$
F	T	$\frac{3}{16}$
F	F	$\frac{5}{16}$

Here's what the Bayesian evolution of this prior looks like, first at t_0 (before anything is learned), then at t_1 (when \mathcal{E} is learned), and finally at t_2 (when \mathcal{A} is learned).

$In[*]:=$ EvaluateProbability[{Pr[$\mathcal{A} \wedge \mathcal{E} \wedge \mathcal{H}$], Pr[$\mathcal{A} \wedge \mathcal{E} \wedge \neg \mathcal{H}$], Pr[$\mathcal{A} \wedge \neg \mathcal{E} \wedge \mathcal{H}$], Pr[$\mathcal{A} \wedge \neg \mathcal{E} \wedge \neg \mathcal{H}$], Pr[$\neg \mathcal{A} \wedge \mathcal{E} \wedge \mathcal{H}$], Pr[$\neg \mathcal{A} \wedge \mathcal{E} \wedge \neg \mathcal{H}$], Pr[$\neg \mathcal{A} \wedge \neg \mathcal{E} \wedge \mathcal{H}$], Pr[$\neg \mathcal{A} \wedge \neg \mathcal{E} \wedge \neg \mathcal{H}$]}, prior]
 $Out[*]:=$ $\left\{ \frac{7}{32}, \frac{1}{32}, \frac{1}{32}, \frac{7}{32}, \frac{3}{32}, \frac{5}{32}, \frac{5}{32}, \frac{3}{32} \right\}$

$In[*]:=$ EvaluateProbability[{Pr[$\mathcal{A} \wedge \mathcal{E} \wedge \mathcal{H} \mid \mathcal{E}$], Pr[$\mathcal{A} \wedge \mathcal{E} \wedge \neg \mathcal{H} \mid \mathcal{E}$], Pr[$\mathcal{A} \wedge \neg \mathcal{E} \wedge \mathcal{H} \mid \mathcal{E}$], Pr[$\mathcal{A} \wedge \neg \mathcal{E} \wedge \neg \mathcal{H} \mid \mathcal{E}$], Pr[$\neg \mathcal{A} \wedge \mathcal{E} \wedge \mathcal{H} \mid \mathcal{E}$], Pr[$\neg \mathcal{A} \wedge \mathcal{E} \wedge \neg \mathcal{H} \mid \mathcal{E}$], Pr[$\neg \mathcal{A} \wedge \neg \mathcal{E} \wedge \mathcal{H} \mid \mathcal{E}$], Pr[$\neg \mathcal{A} \wedge \neg \mathcal{E} \wedge \neg \mathcal{H} \mid \mathcal{E}$]}, prior]
 $Out[*]:=$ $\left\{ \frac{7}{16}, \frac{1}{16}, 0, 0, \frac{3}{16}, \frac{5}{16}, 0, 0 \right\}$

```
In[*]:= EvaluateProbability[{Pr[A ∧ E ∧ H | E ∧ A], Pr[A ∧ E ∧ ¬H | E ∧ A],
  Pr[A ∧ ¬E ∧ H | E ∧ A], Pr[A ∧ ¬E ∧ ¬H | E ∧ A], Pr[¬A ∧ E ∧ H | E ∧ A],
  Pr[¬A ∧ E ∧ ¬H | E ∧ A], Pr[¬A ∧ ¬E ∧ H | E ∧ A], Pr[¬A ∧ ¬E ∧ ¬H | E ∧ A]}, prior]
```

Out[*]=

$\left\{ \frac{7}{8}, \frac{1}{8}, 0, 0, 0, 0, 0, 0 \right\}$

A	E	H	cr _{t₀}	cr _{t₁}	cr _{t₂}
T	T	T	$\frac{7}{32}$	$\frac{7}{16}$	$\frac{7}{8}$
T	T	F	$\frac{1}{32}$	$\frac{1}{16}$	$\frac{1}{8}$
T	F	T	$\frac{1}{32}$	0	0
T	F	F	$\frac{7}{32}$	0	0
F	T	T	$\frac{3}{32}$	$\frac{3}{16}$	0
F	T	F	$\frac{5}{32}$	$\frac{5}{16}$	0
F	F	T	$\frac{5}{32}$	0	0
F	F	F	$\frac{3}{32}$	0	0

What does Douven’s model of an analogous agent look like? Well, presumably, it starts off the same, but only on the sub-algebra {H,E}, since the agent is not assumed to have A in their algebra of entertained propositions.

E	H	cr _{t₀}
T	T	$\frac{5}{16}$
T	F	$\frac{3}{16}$
F	T	$\frac{3}{16}$
F	F	$\frac{5}{16}$

Here, if the agent were to follow Bayes’s Theorem, then they would assign a value $\frac{5}{8}$ to cr_{t₀}(H | E).

```
In[*]:= EvaluateProbability[Pr[H | E], prior]
```

Out[*]=

$\frac{5}{8}$

Moreover, if we look at the t₂ Bayesian posterior — after learning both E and A, viz., cr_{t₂}(H | E ∧ A) — we have the value $\frac{7}{8}$.

```
In[*]:= EvaluateProbability[Pr[H | E ∧ A], prior]
```

Out[*]=

$\frac{7}{8}$

On the other hand, if the agent follows Douven’s rule for updating on E — assuming that A is true, i.e., assuming c(H,E) > 0 — then we will have:

$$cr_{t_0}(H | E) = \frac{cr_{t_0}(H) cr_{t_0}(E | H) + c(H,E)}{cr_{t_0}(H) cr_{t_0}(E | H) + c(H,E) + cr_{t_0}(\neg H) cr_{t_0}(E | \neg H)}$$

```
In[*]:= DouvenPosterior[c_, pr_] :=
  EvaluateProbability[ $\frac{\text{Pr}[H] \times \text{Pr}[E | H] + c}{\text{Pr}[H] \times \text{Pr}[E | H] + c + \text{Pr}[\neg H] \times \text{Pr}[E | \neg H]}$ , pr]
```

For instance, if $c(H,E) = \frac{1}{10}$, then we will obtain the value $\frac{11}{16}$ for the (Douven) posterior.

```
In[*]:= DouvenPosterior[ $\frac{1}{10}$ , prior]
```

```
Out[*]:=
 $\frac{11}{16}$ 
```

Interestingly, **there is no value of** $c(H,E) \in (0,1)$ such that the Bayesian t_2 posterior $cr_{t_0}(H | E \wedge A)$ is equal to the Douven posterior $cr_{t_0}(H | E)$! Specifically, **only if** $c(H,E) = 1$ will the two models will agree on the t_2 posterior.

```
In[*]:= Reduce[DouvenPosterior[c, prior] ==  $\frac{7}{8}$ , c]
```

```
Out[*]:=
c == 1
```

```
In[*]:= DouvenPosterior[1, prior]
```

```
Out[*]:=
 $\frac{7}{8}$ 
```

Ultimately, these two agents will “look the same” at t_2 — over the $\{H,E\}$ sub-algebra. To wit:

E	H	cr_{t_2}
T	T	$\frac{7}{8}$
T	F	$\frac{1}{8}$
F	T	0
F	F	0

The advantage of the Bayesian approach is that — assuming probabilism — the prior cr_{t_0} *completely determines* what happens in subsequent updates (e.g., at cr_{t_1} and cr_{t_2}). There is no “extra parameter” $c(H,E)$ that has to be fixed in order to determine cr_{t_2} . Also, we don’t need any “exogenous perturbation” to the agent’s credences — it’s just standard conditionalization (first on E and then on A).

Moreover, I would suggest that the best way to fix the value of $c(H,E)$ is by *constructing a full Bayesian hypothetical prior* and using it to “reverse engineer” a value for $c(H,E)$ — as we did above, which led to the value $c(H,E) = 1$. But, in this case, the reverse engineering leads to a value $[c(H,E) = 1]$, which is *outside* the range of values Douven assumes for $c(H,E)$.

In fact, there seems to be *no upper bound* on $c(H,E)$ that can be “reverse engineered” from Bayesian

theory. I have a function for creating models satisfying the above constraints, but which require *arbitrarily large* values of $c(H,E)$.

```
In[*]:= model[c_] := PrSAT[ {
  Pr[H | E] > Pr[H | ¬ E],
  Pr[H | E ∧ A] > Pr[H | E ∧ ¬ A],
  Pr[A | E] == Pr[A | ¬ E],
  Pr[H | A] == Pr[H | ¬ A],
  Pr[H] ==  $\frac{1}{2}$ , Pr[E] ==  $\frac{1}{2}$ , Pr[A] ==  $\frac{1}{2}$ ,
  Pr[H | E ∧ A] ==  $\frac{\text{Pr}[H] \times \text{Pr}[E | H] + c}{\text{Pr}[H] \times \text{Pr}[E | H] + c + \text{Pr}[\neg H] \times \text{Pr}[E | \neg H]}$ 
}, Probabilities → Regular, BypassSearch → True];
```

```
In[*]:= model[100]
```

```
Out[*]=
```

```
{ {A → {a2, a5, a6, a8}, E → {a3, a5, a7, a8}, H → {a4, a6, a7, a8},
  Ω → {a1, a2, a3, a4, a5, a6, a7, a8}}, {a1 →  $\frac{10\,828}{162\,409}$ , a2 →  $\frac{40\,528}{162\,409}$ , a3 →  $\frac{119\,097}{649\,636}$ ,
  a4 →  $\frac{119\,097}{649\,636}$ , a5 →  $\frac{297}{649\,636}$ , a6 →  $\frac{297}{649\,636}$ , a7 →  $\frac{10\,828}{162\,409}$ , a8 →  $\frac{40\,528}{162\,409}$ }}
```